

**Exercice 13-7 : Résolution de l'équation des ondes**


a) Voici le programme correspondant :

```

// Vitesse
c=1;
// Discrétisation en espace
xmin=0; xmax=1; h=0.01; xx=xmin:h:xmax; xx=xx';
npt=length(xx); nx=npt-2; xxint=xx(2:nx+1);
// Matrice de discrétisation en espace 1D
K=2*eye(nx,nx)-diag(ones(nx-1,1),1)
      -diag(ones(nx-1,1),-1);
K1D=K/h^2;
// Discrétisation en temps
Tfin=1; dt=0.9*h/c; ntps=ceil(Tfin/dt);
// Donnée initiale
k=100; x0=(xmin+xmax)/2;
f=exp(-k*(xxint-x0).^2); g=zeros(nx,1);
u0=f; u1=u0+dt*g;
// Résolution en temps – Euler Explicite
for t=1:ntps,
    u=2*u1-c^2*dt^2*K1D*u1-u0;
    u0=u1; u1=u;
    if modulo(t,10)==0
        uu=[0;u;0]; plot2d(xx, uu);
    end
end
end

```


et le résultat au bout du temps  $T = 1$  pour  $c\tau/h = 0,9$ , où  $c$  est la vitesse,  $\tau$  est le pas de temps (appelé  $dt$  dans le programme) et  $h$  le pas d'espace :



On voit l'onde se séparer en deux parties égales, l'une se propage vers la droite avec la vitesse  $c$  et l'autre vers la gauche avec la même vitesse. Une fois qu'elle atteint les bords du domaine, elle se réfléchit avec changement de signe, se propage de nouveau et les deux parties se recombinent au centre du domaine. Il faut noter qu'il y a un changement de

signe pour respecter les conditions aux bords de Dirichlet. En revanche, pour respecter les conditions aux bords de von Neumann, l'onde resterait positive.

Pour un pas de temps  $\tau$  tel que  $c\tau/h = 1,03$  (cas où la condition de stabilité n'est pas satisfaite), on obtient au bout du temps  $T = 1$  :



On remarque qu'ici l'erreur numérique devient de plus en plus grande à cause du manque de stabilité du schéma.


b) Voici le programme correspondant :

```

// Vitesse
c=1;
// Discrétisation en espace
xmin=0; xmax=1; h=0.01; xx=xmin:h:xmax;xx=xx';
npt=length(xx);nx=npt-1;xxint=xx(1:nx);
// Matrice de discrétisation en espace 1D
K=2*eye(nx,nx)-diag(ones(nx-1,1),1)
               -diag(ones(nx-1,1),-1);
K(nx,1)=-1;K(1,nx)=-1;K1D=K/h^2;
// Discrétisation en temps
Tfin=1; dt=0.95*h/c; ntps=ceil(Tfin/dt);
//// Donnée initiale
f=5*sin(2*pi*xxint)+4*sin(4*pi*xxint);g=zeros(nx,1);
u0=f;u1=u0+dt*g;
// Résolution en temps - Euler Explicite
for t=1:ntps,
    u=2*u1-c^2*dt^2*K1D*u1-u0;
    u0=u1; u1=u;
    if modulo(t,10)==0
        uu=[u;u(1)]; plot2d(xx, uu);
    end
end
uu=[u;u(1)]; plot(xx, uu, '*');
uu=[f;f(1)]; plot(xx, uu, 'r');

```

et le résultat pour  $c\tau/h = 0,95$  au bout du temps  $T = 1$  :



La solution initiale (en rouge) coïncide avec la solution au temps  $T = 1$  (en bleu avec étoiles). La période temporelle vaut 1, ce qui est cohérent avec la vitesse de l'onde  $c = 1$  et la longueur du domaine  $L = 1$ . On remarque que les conditions aux bords périodiques empêchent l'onde de se réfléchir aux bords : l'onde qui sort à gauche (respectivement à droite) se retrouve à droite (resp. à gauche).